Maple and Calculus III

The following sections give you some idea of the range of problems that can be solved using Maple. In the first example, we see that Maple can be used very effectively to create an animation that can be saved, then viewed in a web browser. Then we look at more ways to use Maple to plot surfaces, compute limits, derivatives and tangent planes.

The TNB Frame

The following code is available on our course website. We present it to you as an example of two important techniques: (1) How to animate in three dimensions, and (2) how to create a loop in Maple (to repeat a command some given number of times).

```
restart;
with(VectorCalculus): with(plots):
r := t -> < cos(t), sin(t), t >;
Helix:=spacecurve([cos(t),sin(t),t,t=0..2*Pi]):

tnbFrame := TNBFrame(r(t),t):
kappa := Curvature(r(t),t):

Tt := tnbFrame[1]:
Nt := tnbFrame[2]:
Bt := tnbFrame[3]:

oscCt := (cos(theta)/kappa)*Tt+(sin(theta)/kappa)*Nt+(1/kappa)*Nt+r(t):

NumIters := 30;
myAnim:= array(1..NumIters+1):

for N from 0 by 1 to NumIters do
   NN:=(2*Pi/NumIters)*N;
   Circ :=spacecurve(subs(t=NN,oscCt),theta=0..2*Pi,color=black,thickness=2,scaling=cons
   PTt := arrow(r(NN),subs(t=NN,Tt),color=red,   thickness=2):
   PNt := arrow(r(NN),subs(t=NN,Nt),color=yellow,thickness=2):
   PBt := arrow(r(NN),subs(t=NN,Bt),color=green, thickness=2):
   myAnim[N+1] := display3d({Helix,Circ,PTt,PNt,PBt},scaling=constrained):
end do:

display(seq(myAnim[i],i=1..NumIters),insequence=true,scaling=constrained);
```

For extra fun, export the movie as an animated GIF file, and you can open it with Firefox!

Graph the function

Graph $z = f(x, y)$. In this example, $z = \sqrt{9 - x^2 - y^2}$

```
f:=(x,y)->sqrt(9-x^2-y^2);
plot3d(f(x,y),x=-3..3,y=-3..3);
```

Compare that with the following:

```
z:=sqrt(9-x^2-y^2);
plot3d(z,x=-3..3,y=-3..3);
```

Try This:

Plot $z = \sin(xy)$, for $x \in [-10, 10]$ and $y \in [-10, 10]$. Change the color scheme to "Z Hue" by adding the option `shading=zhue`, and increase the number of points used to plot by adding `grid=[80,80]`, and finally remove the grid lines by adding the option `style=surface`

Plot the level curves

Plot the level curves (also called contours) of the function $z = f(x, y)$. The Maple command is `contourplot`

```
with(plots):
z:=(x+y)/(sin(y)+2);
contourplot(z,x=-3..3,y=-3..3);
```

Now change the number of points used and see if the graph changes. We can also tell Maple which contours to graph. In this example, we compare the plot of $f(x, y) = \sin^2(x) + \frac{1}{4}y^2$ with its contours at $1/10, 1/2, 1, 3$- We'll visualize the results in 3-D.

```
g:=(sin(x))^2+(1/4)*y^2;
plot3d(g,x=-5..5,y=-2..2);
contourplot(g,x=-5..5,y=-2..2,contours=[1/10,1/2,1,3]);
contourplot3d(g,x=-5..5,y=-2..2);
```

Multiple Limits.

Examples:

Compute the limits (if they exist) in Maple:

$$\lim_{(x,y)\to(0,0)} \frac{x^2 - y^2}{x^2 + y^2} \qquad \lim_{(x,y)\to(0,0)} \frac{3x^2 y}{x^2 + y^2}$$

```
limit( (x^2-y^2)/(x^2+y^2), {x=2,y=1} );
limit( (x^2-y^2)/(x^2+y^2), {x=0,y=0} );
limit( (3*x^2*y)/(x^2+y^2), {x=0,y=0} );
```

Maple could not compute the second (or third) limit- Try graphing them to see if the limit exists at the origin:

```
plot3d((x^2-y^2)/(x^2+y^2),x=-1..1,y=-1..1);
plot3d(3*x^2*y/(x^2+y^2),x=-1..1,y=-1..1);
```

You should see that the limit does not exist in the first graph, the limit is zero in the second (even though Maple could not compute it).

Partial Derivatives:

Before Going Any Further: If you've been following the examples in a Maple worksheet, you might remove all the output

```
Edit -> Remove Output -> From Worksheet
```

save the worksheet and clear the variables before going further:

```
restart;
```

Example:

If $f(x, y, z) = xe^{xy} \ln(z)$. Compute some first and second partial derivatives:

```
f:=x*exp(x*y)*ln(z);
fx:=diff(f,x);
fxz:=diff(fx,z);
fxx:=diff(f,x$2);
fyz:=diff( diff(f,y),z);
```

Notice the `$2` gives you the second derivative.

Example:

Use the definition of the partial derivative to compute $f_x(x, y, z)$, if

$$f(x, y, z) = \sqrt{x^2 + y^2 + z^2}$$

```
f:=(x,y,z)->sqrt(x^2+y^2+z^2);
DiffQuot:=(f(x+h,y,z)-f(x,y,z))/h;
fx:=limit(DiffQuot,h=0);
```

The Gradient

Compute the Gradient, $\nabla f = [f_x, f_y, f_z]$
In Maple, if $f(x,y,z) = 3x^2 + 2yz + 5x - 6y$, then the gradient is computed as the following. We will also find where the gradient is zero, and check by substituting it back into the gradient, and for fun, substitute it into the function:

```
with(linalg):
f:=3*x^2+2*y*z-5*x+6*y;
df:=grad(f, vector([x,y,z]));
S:=solve({df[1]=0,df[2]=0,df[3]=0});
subs(S,[df[1],df[2],df[3]]);
subs(S,f);
```

Contours and Gradients

Plot the contours of $f(x,y) = x^2 - y^2$, together with some gradient vectors.

```
with(plots):
A:=contourplot(x^2-y^2,x=-4..4,y=-4..4):
B:=gradplot(x^2-y^2,x=-4..4,y=-4..4,grid=[6,6],arrows='slim'):
display({A,B});
```

Example In Depth:

The length of a diagonal of a box is to be 1 meter. Find the maximum possible volume.
Solution: Let x, y be the dimensions of the base of the box, and z be its height. We want to find the maximum of $V = xyz$, where there is a restriction on the diagonal- Its length is 1:

$$V = xyz \qquad \text{where } \sqrt{x^2 + y^2 + z^2} = 1$$

Solving the restriction for z gives a function of two variables- Find the maximum of

$$V = xy\sqrt{1 - x^2 - y^2} \quad \text{where } x^2 + y^2 \le 1$$

The minimum and the maximum of V occurs at either its critical points or along the boundary, so we check both:

```
with(linalg):
V:=x*y*sqrt(1-x^2-y^2);
dV:=grad(V,vector([x,y]));
H1:=[solve({dV[1]=0,dV[2]=0})];
H2:=[allvalues(H1[2])];
H3:=subs(H2[1],V);
```

Now, $V = 0$ along the circle $x^2 + y^2 = 1$, and along $x = 0$ or $y = 0$. Therefore, the maximum value is in `H3`, and it occurs at the critical point in `H2[1]` (which was $x = y = 1/\sqrt{3}$)